
Come iniziare a scrivere un programma in ASSEMBLER

Quando si parte a scrivere un programma in assembler bisogna tenere presenti alcune regole basilari che qui vengono riportate.

Tutte le direttive per l'assemblatore (W-ON, PP-ON ecc) vanno messe all'inizio del programma.

Tutti i file di inclusione vanno posizionati subito dopo le direttive dell'assemblatore

Eventuali altre definizioni di costanti o variabili vanno messe subito dopo. 

Ricordatevi di inizare scrivendo da sinistra, senza lasciare spazi e spaziare usando TAB (vedi sotto)

pippo
.equ
7

Vanno sempre definiti i vettori di Interrupt che vanno messi alla fine del programma.

L'ultima istruzione del programma deve essere END
Tutte le LABEL devono essere al massimo di 8 caratteri. Non si deve definire una label con due valori differenti, in questo caso l'ultima definizione e' quella che vale.

E' consigliabile usare la sequenza


AST6 (assemblatore) e LST5 (linker) 

perche' da piu' flessibbilita' e trasportabilita' ai programmi scritti, in questo caso NON si usano le direttive ORG ma le SECTION (vedi piu' avanti). Sicuramente quando si superano i 4K si deve usare la sequenza sopra ricordata.

Quando si usa AST6 e LST6 si usino le seguenti sintassi:

ast6 -l -e -o nome_file

lst6 @memoria.txt -x... nome_file nome_file

Nel file memoria.txt si definiranno le aree di memoria del micro, vedi esempio sotto.
-P0:80-7FF -P1:800-F9F -P32:FF0-FFF -P2:1010-17FF -P3:1810-1FFF
File di esempio in cui si fa uso delle SECTION

;|===========================================================================|

;|                          Assembly directives                              |

;|===========================================================================|

         .title "C0"            ; This directive defines a new title to be

                                ; in the page header of the output listing.

         .vers "ST6230"

        .romsize 8

        .W_ON

        .PP_ON

        .DP_ON

;*****************************************************************************

;*                              data space                                   *

;*****************************************************************************

;|===========================================================================|

;|                           data registers                                  |

;|===========================================================================|

        .PAGE_D 0

         .input "6215_reg.asm"

;|===========================================================================|

;|                            constants definition                           |

;|===========================================================================|

CopiA   .def    84h             ;copia porta a

CopiB   .def    85h             ;copia porta b

CopiD   .def    86h             ;copia porta d

digit1  .def    87h             ;digit1 bit0=>com (deve essere sempre 0)

digit2  .def    88h             ;digits

digit3  .def    89h             ;digits

digit4  .def    8ah             ;digits

digit5  .def    8bh             ;digits

digit6  .def    8ch             ;digits

output  .def    8dh             ;uscite 4094

sblink  .def    8eh             ;stato blink

cblink  .def    8fh             ;counter per blink

blink   .def    90h             ;8 bit blink digit

fkey1   .def    91h             ;8 bit tasti

fkey2   .def    92h             ;4 bit bassi tasti

rxser   .def    93h             ;dato rx su seriale

temp1   .def    96h             ;temp per rti

temp2   .def    97h             ;temp per rti

temp3   .def    98h             ;temp per rti

temp4   .def    99h             ;temp per rti

temp5   .def    9ah             ;temp per rti

temp6   .def    9bh             ;temp per rti

temp7   .def    9ch             ;temp per rti

key1    .def   09dh             ;gestione tasti

key2    .def   09eh             ;gestione tasti

key1p   .def   09fh             ;gestione tasti

key2p   .def   0a0h             ;gestione tasti

stser   .def   0a1h             ;stato seriale

tdgt1   .def   0a9h             ;temp per disp on off

tdgt2   .def   0aah             ;temp per disp on off

tdgt3   .def   0abh             ;temp per disp on off

tdgt4   .def   0ach             ;temp per disp on off

tdgt5   .def   0adh             ;temp per disp on off

tdgt6   .def   0aeh             ;temp per disp on off

time    .equ    34h             ;con prescaler /128 =10ms

bst     .equ    4               ;BIT STROBE 4094

bck     .equ    6               ;bit clock 4094

bdata   .equ    5               ;bit data 4094

bf1     .equ    5               ;bit freccia 1

bf2     .equ    4               ;bit freccia2

bbl     .equ    3               ;bit backlight lcd

bbuz    .equ    2               ;bit buzzer

bl2     .equ    1               ;bit L2

bl1     .equ    0               ;bit L1

reflcd  .equ    7               ;bit flag refresh lcd

bdser   .equ    6               ;bit direzione seriale asinc.

rxrdy   .equ    7               ;bit rx ready asinc.

txmt    .equ    6               ;bit tx empty  "

txstart .equ    0ffh            ;byte start risposta

tmz     .equ    7               ;bit tmz tscr

tblink  .equ    25              ;tempo lampeggio 1=20mS

;***MACROISTRUZIONI***

        .SECTION 0

        nop

        .SECTION 1

;--OutA--  fuori sul port A la sua copia

        .macro OutA

        ld     a,CopiA

        ld     dra,a

        .endm

;--OutB-- fuori sul port B la sua copia

        .macro OutB

        ld     a,CopiB

        ld     drb,a

        .endm

;--OutD-- fuori sul port d la sua copia

        .macro OutD

        ld     a,CopiD

        ld     drd,a

        .endm

;--mclk -- macro clock 4094

        .macro mclk 

        set    bck,CopiB

        OutB

        res    bck,CopiB

        OutB

        .endm

;--mbuzon -- macro buzzer on

        .macro mbuzon

        set    bbuz,output

        .endm

;--mbuzoff -- macro buzzer off

        .macro mbuzoff

        res    bbuz,output

        .endm

;--mbblon -- macro backlight lcd on

        .macro mbblon

        set    bbl,output

        .endm

;--mbbloff -- macro backlight lcd on

        .macro mf1off

        res    bbl,output

        .endm

;--mf1on -- macro backlight lcd on

        .macro mf1on

        set    bf1,output

        .endm

;--mf1off -- macro backlight lcd on

        .macro mbbloff

        res    bf1,output

        .endm

;--mf2on -- macro backlight lcd on

        .macro mf2on

        set    bf2,output

        .endm

;--mf2off -- macro backlight lcd on

        .macro mf2off

        res    bf2,output

        .endm

;--moutbit -- macro data 4094

        .macro moutbit nbit,reg,?lp

         set bdata,CopiB        ;set data

         jrs nbit,reg,lp

         res bdata,CopiB

lp       OutB

         mclk                   ;clock

        .endm

;--moutbyte - macro data 4094

        .macro moutbyte nbyte,?lps

         moutbit 7,nbyte,lps'7

         moutbit 6,nbyte,lps'6

         moutbit 5,nbyte,lps'5

         moutbit 4,nbyte,lps'4

         moutbit 3,nbyte,lps'3

         moutbit 2,nbyte,lps'2

         moutbit 1,nbyte,lps'1

         moutbit 0,nbyte,lps'0

        .endm

;*****************************************************************************

;*                             PROGRAM SPACE                                 *

;*****************************************************************************

;|===========================================================================|

;|                             main program                                  |

;|===========================================================================|

;--moutbyte - macro data 4094

mob:        ;macro moutbyte nbyte,?lps

         moutbit 7,w,lps7

         moutbit 6,w,lps6

         moutbit 5,w,lps5

         moutbit 4,w,lps4

         moutbit 3,w,lps3

         moutbit 2,w,lps2

         moutbit 1,w,lps1

         moutbit 0,w,lps0

         ret  ;endm

mwob:

        ld   a,temp7

        ld   w,a

        call mob

        ld   a,temp6

        ld   w,a

        call mob

        ld   a,temp5

        ld   w,a

        call mob

        ld   a,temp4

        ld   w,a

        call mob

        ld   a,temp3

        ld   w,a

        call mob

        ld   a,temp2

        ld   w,a

        call mob

        ld   a,temp1

        ld   w,a

        call mob

        ret

;**************************************************************************

;*  inizializzazione i/o

;**************************************************************************

reset    ldi ddra,#00010000b    ;pa5 = col0

                                ;pa4 = strobe 4094

         ldi ora,#00010000b     ;pa3 = col1

                                ;pa2 = col2

         ldi dra,#00000000b     ;pa1 = col3

         ldi ddrb,#01100000b    ;pb6 = clock 4094

         ldi orb,#01100000b     ;pb5= data 4094

                                ;pb4= col 4

         ldi ddrd,#01100000b    ;pd7 = col 5

         ldi ord,#01100000b     ;pd6= direzione 75176

         ldi drd,#00000000b     ;pd5= tx data

                                ;pd4= rx data

         ldi uartcr,#00110101b  ; 4800 baud

         reti

         ldi ior,#010h          ;abilita interrupt

         clr digit1             ;init variabili                

         clr digit2                

         clr digit3

         clr digit4

         clr digit5

         clr digit6

         ldi output,#11000000b       ;set out

         clr stser

         clr sblink

         ldi cblink,#004h

         clr blink

         clr CopiA

         clr CopiB

         mbuzoff                ;spegne buzzer

         mbblon                 ;accende retro illuminazione lcd

         res bdser,CopiD

         OutD

mloop:

;***********************************************************************

;***********************************************************************

;******  refresh lcd non complementato

;***********************************************************************

;***********************************************************************

         res bst,CopiA          ;reset strobe

         OutA

         ld  a,output

         ld  temp7,a

         ld  a,output

         ld  temp7,a

         ld  a,digit6

         ld  temp6,a

         ld  a,digit5

         ld  temp5,a

         ld  a,digit4

         ld  temp4,a

         ld  a,digit3

         ld  temp3,a

         ld  a,digit2

         ld  temp2,a

         ld  a,digit1

         ld  temp1,a

         call mwob

         set bst,CopiA          ;set strobe

         OutA

         res bst,CopiA          ;reset strobe

         OutA

;***********************************************************************

;***********************************************************************

;*********  I scan tastiera

;***********************************************************************

;***********************************************************************

         ld   a,key1

         ld   key1p,a

         ld   a,key2

         ld   key2p,a

sk       res  0,key1

         res  1,key1

         res  2,key1

         res  3,key1

         res  4,key1

         res  5,key1

         jrr  5,dra,ska

         set  0,key1

ska      jrr  3,dra,skb

         set  1,key1

skb      jrr  2,dra,skc

         set  2,key1

skc      jrr  1,dra,skd

         set  3,key1

skd      jrr  4,drb,ske

         set  4,key1

ske      jrr  7,drd,skf

         set  5,key1

skf      res  7,output                  ;setta bit per scan successivo

         set  6,output

;***********************************************************************

;***********************************************************************

;******  delay

;***********************************************************************

;***********************************************************************

         ldi  x,#02h

del1     ldi  a,#0ffh

del2     dec  a

         jrnz del2

         dec  x

         jrnz del1

;***********************************************************************

;***********************************************************************

;******  refresh lcd  complementato   

;***********************************************************************

;***********************************************************************

         ld  a,output

         jrs 0,a,rti2a

         set 0,a

         jp  rti2b

rti2a    res 0,a

rti2b    jrs 1,a,rti2c

         set 1,a

         jp  rti2d

rti2c    res 1,a

rti2d:

         ld  temp7,a

         ld  a,digit6

         com a

         ld  temp6,a

         ld  a,digit5

         com a

         ld  temp5,a

         ld  a,digit4

         com a

         ld  temp4,a

         ld  a,digit3

         com a

         ld  temp3,a

         ld  a,digit2

         com a

         ld temp2,a

         ld  a,digit1

         com a

         ld  temp1,a

         call mwob

rti3     set bst,CopiA          ;set strobe

         OutA

         res bst,CopiA          ;reset strobe

         OutA

;***********************************************************************

;***********************************************************************

;*********  II scan tastiera

;***********************************************************************

;***********************************************************************

sk1:     res  6,key1

         res  7,key1

         res  0,key2

         res  1,key2

         res  2,key2

         res  3,key2

         jrr  5,dra,sk1a

         set  6,key1

sk1a     jrr  3,dra,sk1b

         set  7,key1

sk1b     jrr  2,dra,sk1c

         set  0,key2

sk1c     jrr  1,dra,sk1d

         set  1,key2

sk1d     jrr  4,drb,sk1e

         set  2,key2

sk1e     jrr  7,drd,sk1f

         set  3,key2

sk1f     res  6,output                  ;setta bit per scan successivo

         set  7,output

;***********************************************************************

;***********************************************************************

;******  FILTRO TASTIERA: CAMBIA IL VALORE DI FKEY1 E FKEY 2 SOLO SE

;******  LE ULTIME 2 LETTURE SONO IDENTICHE

;***********************************************************************

;***********************************************************************

rdkbd0:                                 

        jrr  0,key1,rdkbd1

        jrr  0,key1p,rdkbd2

        set  0,fkey1

        jp   rdkbd2

rdkbd1: jrs  0,key1p,rdkbd2

        res  0,fkey1

rdkbd2:

        jrr  1,key1,rdkbd3

        jrr  1,key1p,rdkbd4

        set  1,fkey1

        jp   rdkbd4

rdkbd3: jrs  1,key1p,rdkbd4

        res  1,fkey1

rdkbd4:

        jrr  2,key1,rdkbd5

        jrr  2,key1p,rdkbd6

        set  2,fkey1

        jp   rdkbd6

rdkbd5: jrs  2,key1p,rdkbd6

        res  2,fkey1

rdkbd6:

        jrr  3,key1,rdkbd7

        jrr  3,key1p,rdkbd8

        set  3,fkey1

        jp   rdkbd8

rdkbd7: jrs  3,key1p,rdkbd8

        res  3,fkey1

rdkbd8:

        jrr  4,key1,rdkbd9

        jrr  4,key1p,rdkbd10

        set  4,fkey1

        jp   rdkbd10

rdkbd9: jrs  4,key1p,rdkbd10

        res  4,fkey1

rdkbd10:

        jrr  5,key1,rdkbd11

        jrr  5,key1p,rdkbd12

        set  5,fkey1

        jp   rdkbd12

rdkbd11:jrs  5,key1p,rdkbd12

        res  5,fkey1

rdkbd12:

        jrr  6,key1,rdkbd13

        jrr  6,key1p,rdkbd14

        set  6,fkey1

        jp   rdkbd14

rdkbd13:jrs  6,key1p,rdkbd14

        res  6,fkey1

rdkbd14:

        jrr  7,key1,rdkbd15

        jrr  7,key1p,rdkbd16

        set  7,fkey1

        jp   rdkbd16

rdkbd15:jrs  7,key1p,rdkbd16

        res  7,fkey1

rdkbd16:

        jrr  0,key2,rdkbd17

        jrr  0,key2p,rdkbd18

        set  0,fkey2

        jp   rdkbd18

rdkbd17:jrs  0,key2p,rdkbd18

        res  0,fkey2

rdkbd18:

        jrr  1,key2,rdkbd19

        jrr  1,key2p,rdkbd20

        set  1,fkey2

        jp   rdkbd20

rdkbd19:jrs  1,key2p,rdkbd20

        res  1,fkey2

rdkbd20:

        jrr  2,key2,rdkbd21

        jrr  2,key2p,rdkbd22

        set  2,fkey2

        jp   rdkbd22

rdkbd21:jrs  2,key2p,rdkbd22

        res  2,fkey2

rdkbd22:

        jrr  3,key2,rdkbd23

        jrr  3,key2p,rdkbd24

        set  3,fkey2

        jp   rdkbd24

rdkbd23:jrs  3,key2p,rdkbd24

        res  3,fkey2

rdkbd24:

;***********************************************************************

;***********************************************************************

;******  delay

;***********************************************************************

;***********************************************************************

         ldi  x,#02h

del3     ldi  a,#0ffh

del4     dec  a

         jrnz del4

         dec  x

         jrnz del3

;***********************************************************************

;***********************************************************************

;******  blink

;***********************************************************************

;***********************************************************************

        dec  cblink                       ;decrementa contatore x blink

        jrz  blink1

        jp   endbl

blink1:

        ldi  cblink,#tblink               ;ricarica il contatore

        ld   a,sblink

        jrnz blink2     

        jp   bl3

blink2  clr  sblink

        jrr  0,blink,bl20

        ld   a,digit1

        ld   tdgt1,a

        clr  digit1

bl20    jrr  1,blink,bl21

        ld   a,digit2

        ld   tdgt2,a

        clr  digit2

bl21    jrr  2,blink,bl22

        ld   a,digit3

        ld   tdgt3,a

        clr  digit3

bl22    jrr  3,blink,bl23

        ld   a,digit4

        ld   tdgt4,a

        clr  digit4

bl23    jrr  4,blink,bl24

        ld   a,digit5

        ld   tdgt5,a

        clr  digit5

bl24    jrr  5,blink,bl25

        ld   a,digit6

        ld   tdgt6,a

        clr  digit6

bl25    jrr  6,blink,bl26               ;blink go

        ld   a,digit6

        ld   tdgt6,a

        andi a,#0feh                    ;spegne solo go

        ld   digit6,a

bl26:

        jp   endbl

bl3     ldi  sblink,#01h

        jrr  0,blink,bl30

        ld   a,tdgt1

        ld   digit1,a

bl30    jrr  1,blink,bl31

        ld   a,tdgt2

        ld   digit2,a

bl31    jrr  2,blink,bl32

        ld   a,tdgt3

        ld   digit3,a

bl32    jrr  3,blink,bl33

        ld   a,tdgt4

        ld   digit4,a

bl33    jrr  4,blink,bl34

        ld   a,tdgt5

        ld   digit5,a

bl34    jrr  5,blink,bl35

        ld   a,tdgt6

        ld   digit6,a

bl35    jrr  6,blink,bl36       ;recupera solo segmento go

        ld   a,tdgt6

        ld   digit6,a

bl36:

        jp   endbl

endbl:        

        jp   mloop

ciclo principale

;|================================================================|

;|                          interrupt routine                     |

;|================================================================|

seriale:

         ld  v,a                        ;salva accumulatore

gestser:

         jrr rxrdy,uartcr,ser           ;gestione seriale

         res rxrdy,uartcr

ser:     jrr txmt,uartcr,ser1           ;gestione seriale

         res txmt,uartcr

ser1:

         ld  a,uartdr

         ld  rxser,a

         ld  a,stser

         jrz st0

         jp  sercmd

st0:    ld   a,rxser

        cpi  a,#81h             ;rx ser code 81: rx byte digit 1 lcd

        jrnz mlp1

        ldi  stser,#001h        ;prossimo stato seriale = 1

        jp   serend

mlp1    cpi  a,#82h             ;rx ser code 82: rx byte digit 2 lcd

        jrnz mlp2

        ldi  stser,#002h        ;prossimo stato seriale = 2

        jp   serend

mlp2    cpi  a,#83h             ;rx ser code 83: rx byte digit 3 lcd

        jrnz mlp3

        ldi  stser,#003h        ;prossimo stato seriale = 3

        jp   serend

mlp3    cpi  a,#84h             ;rx ser code 84: rx byte digit 4 lcd

        jrnz mlp4 

        ldi  stser,#004h        ;prossimo stato seriale = 4

        jp   serend

mlp4    cpi  a,#85h             ;rx ser code 85: rx byte digit 5 lcd

        jrnz mlp5 

        ldi  stser,#005h        ;prossimo stato seriale = 5

        jp   serend

mlp5    cpi  a,#86h             ;rx ser code 86: rx byte digit 6 lcd

        jrnz mlp6 

        ldi  stser,#006h        ;prossimo stato seriale = 6

        jp   serend

mlp6:    

        cpi  a,#90h             ;rx ser code 90: read keyboard

        jrz  mlp6a 

        jp   mlp7

mlp6a:

        set bdser,CopiD         ;direzione SN75176

        OutD

endrx0: ;jrr txmt,uartcr,endrx0  ;aspetta tx libero NO! si ferma qui(probabilmente viene influenza da rx)

        ldi a,#txstart          ;trasmette start risposta

        ld  uartdr,a

        ldi stser,#07           ;prossimo stato seriale = 7

        jp  serend

mlp7:

        cpi  a,#98h             ;rx ser code 98: disp off

        jrnz mlp8 

        ldi  stser,#008h        ;jp   dispoff

        jp   serend

mlp8:

        cpi  a,#99h             ;rx ser code 98: disp on

        jrnz mlp9 

        ldi  stser,#009h        ;jp   dispon

        jp   serend

mlp9:

        cpi  a,#0a1h            ;rx ser code a1: rx byte tdgt1

        jrnz mlp10

        ldi  stser,#00ah        ;jp   rxtdg1

        jp   serend

mlp10:

        cpi  a,#0a2h            ;rx ser code a2: rx byte tdgt2

        jrnz mlp11

        ldi  stser,#00bh        ;jp   rxtdg2

        jp   serend

mlp11:

        cpi  a,#0a3h            ;rx ser code a3: rx byte tdgt3

        jrnz mlp12

        ldi  stser,#00ch        ;jp   rxtdg3

        jp   serend

mlp12:

        cpi  a,#0a4h            ;rx ser code a4: rx byte tdgt4

        jrnz mlp13

        ldi  stser,#00dh        ;jp   rxtdg4

        jp   serend

mlp13:

        cpi  a,#0a5h            ;rx ser code a5: rx byte tdgt5

        jrnz mlp14

        ldi  stser,#00eh        ;jp   rxtdg5

        jp   serend

mlp14:

        cpi  a,#0a6h            ;rx ser code a6: rx byte tdgt6

        jrnz mlp15

        ldi  stser,#00fh        ;jp   rxtdg6

        jp   serend

mlp15:

        cpi  a,#0b0h            ;rx ser code b0: rx byte per blinking

        jrnz mlp16

        ldi  stser,#00ah        ;stato read blink

        jp   serend

mlp16:

        cpi  a,#0b1h            ;rx ser code b1: rx byte per uscite

        jrnz mlp17

        ldi  stser,#00bh        ;stato read outs

        jp   serend

mlp17:  jp   serend

sercmd:                         ;stati seriale

        cpi  a,#001h            ;

        jrz  cmd1

        jp   cmd2

cmd1    ld   a,rxser            ;rx digit 1

        andi a,#07fh            ;fa and logico (bit 7 e` il comune dell LCD)

        ld   digit1,a           ;carica digit 1

        ld   tdgt1,a

        jp   serenc

cmd2    cpi  a,#002h

        jrz  cmd3

        jp   cmd4

cmd3    ld   a,rxser            ;rx digit 2

        ld   digit2,a           ;carica digit 1

        ld   tdgt2,a

        jp   serenc

cmd4    cpi  a,#003h

        jrz  cmd5

        jp   cmd6

cmd5    ld   a,rxser            ;rx digit 3

        ld   digit3,a           ;carica digit 1

        ld   tdgt3,a

        jp   serenc

cmd6    cpi  a,#004h

        jrz  cmd7

        jp   cmd8

cmd7    ld   a,rxser            ;rx digit 4

        ld   digit4,a           ;carica digit 1

        ld   tdgt4,a

        jp   serenc

cmd8    cpi  a,#005h

        jrz  cmd9

        jp   cmd10

cmd9    ld   a,rxser            ;rx digit 5

        ld   digit5,a           ;carica digit 1

        ld   tdgt5,a

        jp   serenc

cmd10   cpi  a,#006h

        jrz  cmd11

        jp   cmd12

cmd11   ld   a,rxser            ;rx digit 6

        ld   digit6,a           ;carica digit 1

        ld   tdgt6,a

        jp   serenc

cmd12   cpi  a,#007h

        jrz  cmd13

        jp   cmd14

cmd13:

endrx1                           ;trasmette fkey 1

        ld  a,fkey1              ;trasmette byte 1 tasti

        com a

        ld  uartdr,a

        ldi stser,#008h         ;prossimo stato seriale 

        jp  serend

cmd14   cpi  a,#008h            ;trasmette fkey 2

        jrz  cmd15

        jp   cmd16

cmd15:

        ld  a,fkey2             ;trasmette byte 2 tasti

        com a

        andi a,#0fh

        ld  uartdr,a

        ldi stser,#009h         ;prossimo stato seriale

        jp  serend

cmd16   cpi  a,#009h            ;disabilita tx 75176

        jrz  cmd17

        jp   cmd18

cmd17:

        res bdser,CopiD

        OutD

        jp   serenc

cmd18   cpi  a,#00ah            ;legge blink

        jrz  cmd19

        jp   cmd20

cmd19:

        ld   a,rxser

        ld   blink,a

        jp   serenc

cmd20   cpi  a,#00bh            ;legge outs

        jrz  cmd21

        jp   cmd22

cmd21:

        ld   a,output

        andi a,#11000000b       ;maschera uscite

        ld   output,a

        ld   a,rxser

        andi a,#03fh            ;maschera bits tastiera

        add  a,output

        ld   output,a

        jp   serenc

cmd22:

serenc  clr  stser

serend:

         ld  a,v                       ;riprende accumulatore

         reti

;*****************************************************************

;*                     Restart and interrupt Vectors             *

;*****************************************************************

        .SECTION 32

source4  jp   seriale

source3  nop

         reti

source2  nop

         reti

source1  nop

         reti

source0  nop

         reti

         nop

         nop

nmi      nop

         reti

res      jp reset

        .END


